
The key to any nameless representation of syntax is how it indicates the variables we choose to use

and thus, implicitly, those we discard. Standard de Bruijn representations delay discarding maxi-

mally till the leaves of terms where one is chosen from the variables in scope at the expense of the

rest. Consequently, introducing new but unused variables requires term traversal. This paper intro-

duces a nameless ‘co-de-Bruijn’ representation which makes the opposite canonical choice, delaying

discarding minimally, as near as possible to the root. It is literate Agda: dependent types make

it a practical joy to express and be driven by strong intrinsic invariants which ensure that scope is

aggressively whittled down to just the support of each subterm, in which every remaining variable

occurs somewhere. The construction is generic, delivering a universe of syntaxes with higher-order

metavariables, for which the appropriate notion of substitution is hereditary. The implementation

of simultaneous substitution exploits tight scope control to avoid busywork and shift terms without

traversal. Surprisingly, it is also intrinsically terminating, by structural recursion alone.

When I was sixteen and too clever by half, I wrote a text editor which cached a plethora of useful

but redundant pointers into the buffer, just to shave a handful of nanoseconds off redisplay. Accurately

updating these pointers at each keystroke was a challenge which taught me the hard way about the value

of simplicity. Now, I am a dependently typed programmer. I do not keep invariants: invariants keep me.

This paper is about scope invariants in nameless representations of syntax. One motivation for such is

eliminating redundant name choice to make α-equivalence trivial. Classic de Bruijn syntaxes [9] replace

name by number: variable uses count either out from use to binding (indices), or in from root to binding

(levels). Uses are found at the leaves of syntax trees, so any operation which modifies the sequence of

variables in scope requires traversal. E.g., consider this β -reduction (under λx) in untyped λ -calculus.

name λx. (λy. y x (λ z. z (y z))) (x (λv. v))  β λx. (x (λv. v)) x (λ z. z ((x (λv. v)) z)

index λ . (λ .0 1 (λ .0 (1 0))) (0 (λ .0))  β λ . (0 (λ .0))0 (λ .0 ((1 (λ .0))0)

level λ . (λ .1 0 (λ .2 (1 2))) (0 (λ .1))  β λ . (0 (λ .1))0 (λ .1 ((0 (λ .2))1)

Underlining shows the movement of the substituted term. In the index representation, the free x must be

shifted when it goes under the λ z. With levels, the free x stays 0, but the bound v must be shifted under

λ z, and the substitution context must be shifted to account for the eliminated λy. Shift happens.

The objective of this paper is not to eliminate shifts altogether, but to ensure that they do not require

traversal. The approach is to track exactly which variables are relevant at all nodes in the tree and

aggressively expel those unused in any given subtree. As we do so, we need and obtain much richer

accountancy of variable usage, with much more intricate invariants. Category theory guides the design

of these invariants and Agda’s dependent types [20] drive their correct implementation.

My explorations follow Sato, Pollack, Schwichtenberg and Sakurai, whose λ -terms make binding

sites carry maps of use sites [21]. E.g., the K and S combinators become (respectively)

names λc.λe.c λ f . λ s. λe. ( f e) (s e)
maps 1\ 0\� ((10) (00))\((00) (10))\((01) (01))\ (��) (��)

where each abstraction shows with 1s where in the subsequent tree of applications its variable occurs:

leaves, �, are relieved of choice. Of course, the tree under each binder determines which maps are well

formed in a highly nonlocal way: these invariants are formalised extrinsically both in Isabelle/HOL and

in Minlog, over a context-free datatype enforcing neither scope nor shape. Other prior art along similar

lines includes the Haskell implementation by Abel and Kraus [2] of a similar representation, recording

at each λ which of the variable occurrences free below it are bound by it, in left-to-right order, run-

length encoded. Earlier still, the director strings representation of Kennedy and Sleep, refined by Sinot,

Fernandez and Mackie [16, 22], annotated each node with a mapping from each free variable to the set

of indices of the subnodes in which it occurs, and we shall see something similar here.

However, in this paper, we shall obtain an intrinsically valid representation, enforced by type, where

the map information is localized. Binding sites tell only if the variable is used; the crucial choice points

where a term comprises more than one subterm say which variables go where, as in the director strings

representation. Not all are used in all subterms, but (as Eccles says to Seagoon) everybody’s got to be

somewhere [19]: variables used nowhere have been discarded already. This property is delivered by a

coproduct construction in the slices of the category of order-preserving embeddings, but fear not: we

shall revisit all of the category theory required to develop the definition, especially as it strays beyond

the familiar (e.g., to Haskellers) territory of types-and-functions.

Intrinsically well scoped de Bruijn terms date back to Bellegarde and Hook [6], using option types

to grow a type of free variables, but hampered by lack of polymorphic recursion in ML. Substitution

(i.e., monadic structure) was developed for untyped terms by Bird and Paterson [8] and for simple types

by Altenkirch and Reus [5], both dependent either on a prior implementation of renumbering shifts (i.e.,

functorial structure) or a non-structural recursion. My thesis [17] follows McKinna and Goguen [12] in

restoring a single structural operation abstracting ‘action’ on variables, instantiated to renumbering then

to substitution, an approach subsequently adopted by Benton, Kennedy and Hur [7] and generalised to

semantic actions by Allais et al. [3]. Here, we go directly to substitution: shifts need no traversal.

I present not only λ -calculus but a universe of syntaxes inspired by Harper, Honsell and Plotkin’s

Logical Framework [13]. I lift the sorts of a syntax to higher kinds, acquiring both binding (via subterms

at higher kind) and metavariables (at higher kind). However, substituting a higher-kinded variable de-

mands substitution of its parameters hereditarily [23] and simultaneously. Thereby hangs a tale. Abel

showed how sized types justify this process’s apparently non-structural recursion in MSFP 2006 [1]. As

editor, I anonymised a discussion with a referee which yielded a structural recursion for hereditary sub-

stitution of a single variable, instigating Keller and Altenkirch’s formalization at MSFP 2010 [15]. Here,

at last, simultaneous hereditary substitution becomes structurally recursive.

1 Basic Equipment in Agda

We shall need finite types Zero, One, and Two, named for their cardinality, and the reflection of Two as

a set of evidence for ‘being tt’.

data Zero : Setwhere

record One : Setwhere constructor 〈〉
data Two : Setwhere tt ff : Two

Tt : Two → Set

Tt tt = One

Tt ff = Zero

Dependent pairing is by means of the Σ type, abbreviated by × when non-dependent. The pattern

synonym ! allows the first component to be determined by the second: making it a right-associative

prefix operator lets us write ! ! expression rather than ! (! (expression)).

record Σ (S : Set) (T : S → Set) : Setwhere

constructor ,
field fst : S; snd : T fst

× : Set → Set → Set

S × T = Σ S λ → T

pattern ! t = , t

We shall also need to reason equationally. For all its imperfections in matters of extensionality, it

will be convenient to define equality inductively, enabling the rewrite construct in equational proofs.

data == {X : Set} (x : X) : X → Setwhere refl : x == x

2 ∆
K
+: The (Semi-Simplicial) Category of Order-Preserving Embeddings

No category theorist would mistake me for one of their own. However, the key technology in this paper

can be helpfully conceptualised categorically. Category theory is just the study of compositionality —

for everything, not just sets-and-functions. Here, we have an opportunity to develop categorical structure

away from the usual apparatus for programming with functions. Let us therefore revisit the basics.

Category (I): Objects and Morphisms. A category is given by a class of objects and a family of

morphisms (or arrows) indexed by two objects: source and target. Abstractly, we may write C for a

given category, |C| for its objects, and C(S,T ) for its morphisms with given source and target, S,T ∈ |C|.

The rest will follow, but let us fix these notions for our example category, ∆K
+ , of order-preserving

embeddings between variable scopes. Objects are given as backward (or ‘snoc’) lists of the kinds, K,

of variables. (I habitually suppress K and just write ∆+ for the category.) Backward lists respect the

tradition of writing contexts left of judgements in rules and growing them rightwards. However, I say

‘scope’ rather than ‘context’: we track variable availability, but perhaps not all contextual data. Moreover,

I take the typesetting liberty of hiding inferable prefixes of implicit quantifiers, to reduce clutter.

data Bwd (K : Set) : Setwhere

-, : Bwd K → K → Bwd K

[] : Bwd K

data ⊑ : Bwd K → Bwd K → Setwhere

o′ : iz⊑ jz → iz ⊑ (jz -, k)
os : iz⊑ jz → (iz -, k)⊑ (jz -, k)

oz : [] ⊑ []

The morphisms, iz ⊑ jz, of ∆+ embed a source into a target scope. Colloquially, we may call them

‘thinnings’, as they dilute the variables of the source scope with more. I write step constructors postfix,

so thinnings (like scopes) grow on the right. Now, where I give myself away as a type theorist is that I do

not consider the notion of ‘morphism’ to make sense without prior source and target objects. The type

iz ⊑ jz (which is a little more mnemonic than ∆+(iz, jz)) is the type of ‘thinnings from iz to jz’: there is

no type of ‘thinnings’ per se.

Altenkirch, Hofmann and Streicher [4], from whom I learned this notion, take the dual view of

morphisms as selecting one subcontext from another. When K = One, objects represent numbers and

⊑ generates Pascal’s Triangle; excluding the empty scope and allowing degenerate (non-injective) maps

yields ∆, the simplex category beloved of topologists.

Let us have an example thinning: here, we embed a scope with three variables into a scope with five.

k4 •−−−−−• k4 os : [] -, k0 -, k2 -, k4⊑ [] -, k0 -, k1 -, k2 -, k3 -, k4

◦ k3 o′

k2 •−−−−−• k2 os

◦ k1 o′

k0 •−−−−−• k0 os

oz

Category (II): Identity and Composition. In any category, certain morphisms must exist. Each

object X ∈ |C| has an identity ιX ∈ C(X ,X), and wherever the target of one morphism meets the source

of another, their composite makes a direct path: if f ∈ C(R,S) and g ∈C(S,T ), then ( f ;g) ∈C(R,T ).

E.g., every scope has the identity thinning, oi, and thinnings compose via #. (For functions, it is usual

to write g · f for ‘g after f ’ rather than f ;g for ‘ f then g’, but for thinnings I retain spatial intuition.)

oi : kz⊑ kz

oi {kz = iz -, k} = oi os -- os preserves oi

oi {kz = []} = oz

# : iz⊑ jz → jz⊑ kz → iz⊑ kz

θ # φ o′ = (θ # φ) o′

θ o′ # φ os = (θ # φ) o′

θ os # φ os = (θ # φ) os -- os preserves #
oz # oz = oz

By way of example, let us plot specific uses of identity and composition.

oi θ φ θ # φ

k4 •−−−−−• k4 k4 •−−−−−• k4 k4 •−−−−−• k4 k4 •−−−−−−−−−• k4

k3 •−−−−−• k3 ◦ k3 ◦ k3

k2 •−−−−−• k2 ◦ k2 k2 •−−−−−• k2 ◦ k2

k1 •−−−−−• k1 ◦ k1 ◦ k1

k0 •−−−−−• k0 k0 •−−−−−• k0 k0 •−−−−−• k0 k0 •−−−−−−−−−• k0

Category (III): Laws. To complete the definition of a category, we must say which laws are satisfied

by identity and composition. Composition absorbs identity on the left and on the right. Moreover,

composition is associative, meaning that any sequence of morphisms which fit together target-to-source

can be composed without the specific pairwise grouping choices making a difference. That is, we have

three laws which are presented as equations, at which point any type theorist will want to know what is

meant by ‘equal’: I shall always be careful to say. Our thinnings are first-order, so == will serve. With

this definition in place, we may then state the laws. I omit the proofs, which go by functional induction.

law−oi# : oi # θ == θ law−#oi : θ # oi== θ law−## : θ # (φ # ψ) == (θ # φ) # ψ

As one might expect, order-preserving embeddings have a strong antisymmetry property that one

cannot expect of categories in general. The only invertible arrows are the identities. Note that we must

match on the proof of iz == jz even to claim that θ and φ are the identity.

antisym : (θ : iz⊑ jz) (φ : jz⊑ iz) → Σ (iz == jz) λ {refl → (θ == oi)× (φ == oi)}

Example: de Bruijn Syntax via ∆
One
+ . De Bruijn indices are numbers [9], perhaps a type-enforced

bound [6, 8, 5]. Singleton thinnings, k← kz = [] -, k ⊑ kz, can play this rôle in a syntax.

data Lam (iz : Bwd One) : Setwhere

# : (x : 〈〉 ← iz) → Lam iz

$ : (f s : Lam iz) → Lam iz

λ : (t : Lam (iz -, 〈〉)) → Lam iz

↑ : Lam iz → iz⊑ jz → Lam jz

# i ↑ θ = # (i # θ )
(f $ s) ↑ θ = (f ↑ θ ) $ (s ↑ θ )
λ t ↑ θ = λ (t ↑ θ os)

Variables are represented by pointing, eliminating redundant choice of names, but it is only when we

point to one variable that we exclude the others. Thus de Bruijn indexing effectively uses thinnings to

discard unwanted variables as late as possible, in the leaves of syntax trees.

Note how the scope index iz is the target of a thinning in # and weakened in λ. Hence, thinnings act

on terms ultimately by postcomposition, but because terms keep their thinnings at their leaves, we must

hunt the entire tree to find them. Now consider the other canonical placement of thinnings, nearest the

root, discarding unused variables as early as possible.

3 Slices of Thinnings

If we fix the target of thinnings, ( ⊑ kz), we obtain the notion of subscopes of a given kz. Fixing a target

is a standard way to construct a new category whose objects are given by morphisms of the original.

Slice Category. If C is a category and I one of its objects, the

slice category C/I has as its objects pairs (S, f ), where S is an

object of C and f : S→ I is a morphism in C. A morphism in

(C/I)((S, f ),(T,g)) is some h : S→ T such that f = h;g. (The

dotted regions in the diagram show the objects in the slice.)

S
h //

f
��❂

❂❂
❂❂

❂❂
❂ T

g
����
��
��
��

I

That is, the morphisms are triangles. A seasoned dependently typed programmer will be nervous at

a definition like the following (where the after Σ asks Agda to compute the type iz⊑ jz of θ ):

ψ →/ φ = Σ λ θ → (θ # φ) == ψ -- beware of #!

because the equation restricts us when it comes to manipulating triangles. Dependent pattern matching

relies on unification of indices, but defined functions like # make unification difficult, obliging us to

reason about the edges of the triangles. It helps at this point to define the graph of # inductively.

data Tri : iz⊑ jz → jz⊑ kz → iz⊑ kz → Setwhere

t-′′ : Tri θ φ ψ → Tri θ (φ o′) (ψ o′)
t′s′ : Tri θ φ ψ → Tri (θ o′) (φ os) (ψ o′)
tsss : Tri θ φ ψ → Tri (θ os) (φ os) (ψ os)

tzzz : Tri oz oz oz

tri : (θ : iz⊑ jz) (φ : jz⊑ kz) →
Tri θ φ (θ # φ )

comp : Tri θ φ ψ → ψ == (θ # φ)

The indexing is entirely in constructor form, which will allow easy unification. Moreover, all the data in

a Tri structure come from its indices. Easy inductions show that Tri is precisely the graph of #.

The example composition given above can be rendered a triangle, as follows:

egTri : Tri {kz = [] -, k0 -, k1 -, k2 -, k3 -, k4} (oz os o′ os) (oz os o′ os o′ os) (oz os o′ o′ o′ os)
egTri = tzzz tsss t-′′ t′s′ t-′′ tsss

Morphisms in the slice can now be triangles: ψ →/ φ = Σ λ θ → Tri θ φ ψ .

A useful ∆+-specific property is that morphisms in ∆+/kz are unique. It is easy to state this property

in terms of triangles with common edges, triU : Tri θ φ ψ → Tri θ ′ φ ψ → θ == θ ′, and then prove

it by induction on the triangles, not edges. It is thus cheap to obtain universal properties in the slices of

∆+, asserting the existence of unique morphisms: uniqueness comes for free!

4 A Proliferation of Functors

Haskell makes merry with class Functor and its many subclasses: this scratches but the surface,

giving only endofunctors from types-and-functions to types-and-functions. Once we adopt the general

notion, functoriality sprouts everywhere, with the same structures usefully functorial in many ways.

Functor. A functor is a mapping from a source category C to a target category D which preserves

categorical structure. To specify a structure, we must give a function Fo : |C| → |D| from source objects

to target objects, together with a family of functions Fm : C(S,T )→ D(Fo(S),Fo(T )). The preserved

structure amounts to identity and composition: we must have that Fm(ιX ) = ιFo(X) and that Fm( f ;g) =
Fm( f );Fm(g). Note that there is an identity functor I (whose actions on objects and morphisms are the

identity) from C to itself and that functors compose (componentwise).

E.g., every k : K induces a functor (weakening) from ∆+ to itself by scope extension, ( -, k) on

objects and os on morphisms. The very definitions of oi and # show that os preserves oi and #.

To see more examples, we need more categories. Let Set’s objects be types in Agda’s Set universe

and Set(S,T ) exactly S→ T , with the usual identity and composition. Morphism equality is pointwise.

Exercises: make Bwd : Set → Set a functor; check (Lam,↑) is a functor from ∆+ to Set.

Let us plough a different furrow, rich in dependent types, constructing new categories by indexing.

If I : Set, we may then take I → Set to be the category whose objects are families of objects in Set,

S,T : I → Set with morphisms (implicitly indexed) families of functions: S →̇ T = ∀ {i} → S i → T i.

Morphisms are equal if they map each index to pointwise equal functions. In the sequel, it will be

convenient to abbreviate Bwd K → Set as K, for types indexed over scopes.

Dependently typed programming thus offers us a richer seam of categorical structure than we see in

Haskell. This presents an opportunity to make sense of the categorical taxonomy in terms of concrete

programming examples, and at the same time, organising those programs and indicating what to prove.

5 Things-with-Thinnings (a Monad)

Let us acquire the habit of packing terms together with an object in the slice of thinnings over their scope,

selecting the support of the term and discarding unused variables. Note, ⇑ is a functor from K to itself.

record ⇑ (T : K) (scope : Bwd K) : Setwhere -- (T ⇑ ) : K

constructor ↑
field {support} : Bwd K; thing : T support; thinning : support⊑ scope

map⇑ : (S →̇ T) → ((S ⇑ ) →̇ (T ⇑ ))
map⇑ f (s ↑ θ ) = f s ↑ θ

In fact, the categorical structure of ∆+ makes ⇑ a monad. Let us recall the definition.

Monad. A functor M from C to C gives rise to a monad (M,η ,µ) if we can find a pair of natural

transformations, respectively ‘unit’ (‘add an M layer’) and ‘multiplication’ (‘merge M layers’).

ηX : I(X)→M(X) µX : M(M(X))→M(X)

subject to the conditions that merging an added layer yields the identity (whether the layer added is

‘outer’ or ‘inner’), and that adjacent M layers may be merged pairwise in any order.

ηM(X); µX = ιM(X) M(ηX); µX = ιM(X) µM(X); µX = M(µX); µX

The categorical structure of thinnings makes ⇑ a monad. Here, ‘adding a layer’ amounts to ‘wrapping

with a thinning’. The proof obligations to make (⇑,unit⇑,mult⇑) a monad are exactly those required to

make ∆+ a category in the first place. In particular, things-with-thinnings are easy to thin further, indeed,

parametrically so. In other words, (T ⇑) is uniformly a functor from ∆+ to Set.

unit⇑ : T →̇ (T ⇑ )
unit⇑ t = t ↑ oi

mult⇑ : ((T ⇑ ) ⇑ ) →̇ (T ⇑ )
mult⇑ ((t ↑ θ ) ↑ φ) = t ↑ (θ # φ)

thin⇑ : iz⊑ jz → T ⇑ iz → T ⇑ jz

thin⇑ θ t = mult⇑ (t ↑ θ )

Shortly, we shall learn how to find the variables on which a term syntactically depends. However,

merely allowing a thinning at the root, Lam ⇑ iz, yields a redundant representation, as we may discard

variables at either root or leaves. Let us eliminate redundancy by insisting that a term’s support is

relevant: a variable retained by the thinning must be used in the thing. Everybody’s got to be somewhere.

6 The Curious Case of the Coproduct in Slices of ∆+

The ⇑ construction makes crucial use of objects in the slice category ∆+/scope, which exhibit useful

additional structure: they are bit vectors, with one bit per variable telling whether it has been selected.

Bit vectors inherit Boolean structure, via the ‘Naperian’ array structure of vectors [11].

Initial object. A category C has initial object 0, if there is a unique morphism in C(0,X) for every X .

The empty type is famed for this rôle for types-and-functions: empty case analysis gives the vacu-

ously unique morphism. In ∆+, the empty scope plays this rôle, with the ‘constant 0’ bit vector as unique

morphism. By return of post, we get ([],oe) as the initial object in the slice category ∆+/kz. Hence, we

can make constants with empty support, i.e., noting that no variable is (·R for) relevant.

oe : ∀ {kz : Bwd K} → []⊑ kz

oe {iz -, k} = oe o′

oe {[]} = oz

law−oe : (θ : []⊑ kz) → θ == oe

oe/ : (θ : iz⊑ kz) → oe→/ θ

oe/ θ with tri oe θ

... | t rewrite law−oe (oe # θ ) = oe , t

dataOneR : K where 〈〉 : OneR [] 〈〉R : OneR ⇑ kz; 〈〉R = 〈〉 ↑ oe

We should expect the constant to be the trivial case of some notion of relevant pairing, induced by

coproducts in the slice category. If we have two objects in ∆+/kz representing two subscopes, (iz, θ ) and

(jz, φ), there should be a smallest subscope which includes both: pairwise disjunction of bit vectors.

Coproduct. Objects S and T of category C have a coproduct object S+ T if there are morphisms

l ∈ C(S,S+T ) and r ∈ C(T,S+T ) such that every pair f ∈ C(S,U) and g ∈ C(T,U) factors through a

unique h ∈ C(S+T,U) so that f = l;h and g = r;h. In Set, we may take S+T to be the disjoint union

of S and T , with l and r its injections and h the case analysis whose branches are f and g.

However, we are not working in Set, but in a slice category. Any category theorist will tell you that

slice categories C/I inherit colimit structure (characterized by universal out-arrows) from C, as indeed

we just saw with the initial object. If ∆+ has coproducts, too, we are done! Taking K = One, let us

seek the coproduct of two singletons, S = T = [] -, 〈〉. Construct one diagram by taking U = [] -, 〈〉 and

f = g = oi, ensuring that our only candidate for S+ T is again the singleton [] -, 〈〉, with l = r = oi,

making h = oi. Nothing else can sit between S,T and U .

U

•
U ′

• •

•
S

f

BB✝✝✝✝✝✝✝✝✝✝✝

l
// •
S+T

h

OO

•
T

r
oo

g

\\✾✾✾✾✾✾✾✾✾✾✾

•
S

f ′

BB✝✝✝✝✝✝✝✝✝✝✝

l
// •
S+T

?

��

•
T

r
oo

g′

\\✾✾✾✾✾✾✾✾✾✾✾

Now begin a different diagram, with U ′ = [] -, 〈〉 -, 〈〉, allowing f ′ = oz os o′ and g′ = oz o′ os. No h′

post-composes l and r (both oi, making h′ itself) to yield f ′ and g′ respectively. We do not get coproducts.

Fortunately, we get what we need: ∆+ may not have coproducts, but its slices do. Examine the data:

two subscopes of some kz, θ : iz ⊑ kz and φ : jz ⊑ kz. Their coproduct must be some ψ : ijz ⊑ kz,

where our l and r must be triangles Tri θ ′ ψ θ and Tri φ ′ ψ φ , giving morphisms in θ →/ ψ and φ →/ ψ .

Choose ψ to be pointwise disjunction of θ and φ , minimizing ijz: θ ′ and φ ′ will then cover ijz.

data Cover (ov : Two) : iz⊑ ijz → jz⊑ ijz → Setwhere

c′s : Cover ov θ φ → Cover ov (θ o′) (φ os)
cs′ : Cover ov θ φ → Cover ov (θ os) (φ o′)
css : {both : Tt ov} → Cover ov θ φ → Cover ov (θ os) (φ os)

czz : Cover ov oz oz

The flag, ov, determines whether overlap is permitted: with tt for coproducts and ff for partitions. No

constructor allows both θ and φ to omit a target variable, so everybody’s got to be somewhere. Let us

compute the coproduct, ψ then check that any other diagram for some ψ ′ yields a ψ →/ ψ ′.

cop : (θ : iz⊑ kz) (φ : jz⊑ kz) →

Σ λ ijz → Σ (ijz ⊑ kz) λ ψ →
Σ (iz⊑ ijz) λ θ ′ → Σ (jz⊑ ijz) λ φ ′ →
Tri θ ′ ψ θ × Cover tt θ ′ φ ′ × Tri φ ′ ψ φ

copU : Tri θ ′ ψ θ → Cover tt θ ′ φ ′ → Tri φ ′ ψ φ →
θ →/ ψ ′ → φ →/ ψ ′ → ψ →/ ψ ′

iz

θ

))❚❚❚
❚❚❚

❚❚❚
❚❚❚

❚❚❚
❚❚❚

❚❚❚

θ ′ ��❄
❄

❄
❄

〈 ijz
ψ //❴❴❴❴❴❴ kz

jz

φ

55❦❦❦❦❦❦❦❦❦❦❦❦❦❦❦❦❦❦❦❦❦

φ ′
??⑧

⑧
⑧

⑧

where the 〈 in the diagram indicates that the two incoming arrows form a Cover.

The recursive steps in cop’s implementation work explicitly with the two-dimensional triangles and

coverings, using ! to hide their boundaries (thinnings) and their boundaries’ boundaries (scopes).

cop (θ o′) (φ o′) = let ! ! ! ! tl , c , tr = cop θ φ in ! ! ! ! tl t-′′ , c , tr t-′′

cop (θ o′) (φ os) = let ! ! ! ! tl , c , tr = cop θ φ in ! ! ! ! tl t′s′ , c c′s , tr tsss

cop (θ os) (φ o′) = let ! ! ! ! tl , c , tr = cop θ φ in ! ! ! ! tl tsss , c cs′ , tr t′s′

cop (θ os) (φ os) = let ! ! ! ! tl , c , tr = cop θ φ in ! ! ! ! tl tsss , c css , tr tsss

cop oz oz = ! ! ! ! tzzz , czz , tzzz

The copU proof goes by induction on the triangles which share ψ ′ and inversion of the coproduct.

A further useful property of coproduct diagrams is that we can selectively

refine them by a thinning into the covered scope.

subCop : (ψ : kz⊑ kz′) → Cover ov θ ′ φ ′ →
Σ λ iz → Σ λ jz → Σ (iz⊑ kz) λ θ → Σ (jz⊑ kz) λ φ →
Σ (iz⊑ iz′) λ ψ0 → Σ (jz⊑ jz′) λ ψ1 → Cover ov θ φ

The implementation is a straightforward induction on the diagram.

iz
ψ0 //❴❴❴

θ

��❃
❃

❃
❃ iz′

θ ′

  ❆
❆❆

❆❆
❆❆

❆

〈 kz
ψ //〈 kz′

jz
ψ1

//❴❴❴

φ

@@�
�

�
�

jz′
φ ′

>>⑦⑦⑦⑦⑦⑦⑦⑦

The payoff from coproducts is the type of relevant pairs — the co-de-Bruijn touchstone:

record ×R (S T : K) (ijz : Bwd K) : Setwhere

constructor pair

field outl : S ⇑ ijz; outr : T ⇑ ijz

cover : Cover tt (thinning outl) (thinning outr)

,R : S ⇑ kz → T ⇑ kz → (S ×R T) ⇑ kz

(s ↑ θ ) ,R (t ↑ φ) =
let ! ψ , θ ′ , φ ′ , , c , = cop θ φ

in pair (s ↑ θ ′) (t ↑ φ ′) c ↑ ψ

The corresponding projections are readily definable.

outlR : (S ×R T) ⇑ kz → S ⇑ kz

outlR (pair s ↑ ψ) = thin⇑ ψ s

outrR : (S×R T) ⇑ kz → T ⇑ kz

outrR (pair t ↑ ψ) = thin⇑ ψ t

7 Monoidal Structure of Order-Preserving Embeddings

Variable bindings extend scopes. The λ construct does just one ‘snoc’, but binding can be simultaneous,

so the monoidal structure on ∆+ induced by concatenation is what we need.

++ : Bwd K → Bwd K → Bwd K

kz ++ [] = kz

kz ++ (iz -, j) = (kz ++ iz) -, j

++⊑ : iz⊑ jz → iz′ ⊑ jz′ → (iz ++ iz′)⊑ (jz ++ jz′)
θ ++⊑ oz = θ

θ ++⊑ (φ os) = (θ ++⊑ φ ) os
θ ++⊑ (φ o′) = (θ ++⊑ φ ) o′

Concatenation further extends to Coverings, allowing us to build them in chunks.

++C : Cover ov θ φ → Cover ov θ ′ φ ′ → Cover ov (θ ++⊑ θ ′) (φ ++⊑ φ ′)
c ++C (d c′s) = (c ++C d) c′s
c ++C (d cs′) = (c ++C d) cs′

c ++C ( css {both = b} d) = css {both = b} (c ++C d)
c ++C czz = c

One way to build such a chunk is to observe that two scopes cover their concatenation.

lrCop : (iz jz : Bwd K) → Σ (iz⊑ (iz ++ jz)) λ θ → Σ (jz ⊑ (iz ++ jz)) λ φ → Cover ov θ φ

lrCop iz (jz -, j) = let ! ! c = lrCop iz jz in ! ! c c′s

lrCop (iz -, i) [] = let ! ! c = lrCop iz [] in ! ! c cs′

lrCop [] [] = ! ! czz

Now, crucial to the enterprise is that the monoidal structure of scopes lets us not only combine

thinnings, but split them, into global and local parts.

⊣ : ∀ jz (ψ : iz⊑ (kz ++ jz)) → Σ λ kz′ → Σ λ jz′ →
Σ (kz′ ⊑ kz) λ θ → Σ (jz′ ⊑ jz) λ φ → Σ (iz == (kz′ ++ jz′)) λ {refl → ψ == (θ ++⊑ φ )}

[] ⊣ ψ = ! ! ψ , oz , refl , refl
(jz -, j) ⊣ (ψ os) with jz ⊣ ψ

(jz -, j) ⊣ (. (θ ++⊑ φ) os) | ! ! θ , φ , refl , refl = ! ! θ , φ os , refl , refl
(jz -, j) ⊣ (ψ o′) with jz ⊣ ψ

(jz -, j) ⊣ (. (θ ++⊑ φ) o′) | ! ! θ , φ , refl , refl = ! ! θ , φ o′ , refl , refl

Thus equipped, we can say how to bind some variables. The key is to say at the binding site which of

the bound variables will actually be used: if they are not used, we should not even bring them into scope.

data ⊢ jz (T : K) kz : Setwhere

) : iz⊑ jz → T (kz ++ iz)
→ (jz ⊢ T) kz

)R : ∀ jz → T ⇑ (kz ++ jz) → (jz ⊢ T) ⇑ kz

jz )R (t ↑ ψ) with jz ⊣ ψ

jz )R (t ↑ . (θ ++⊑ φ )) | ! ! θ , φ , refl , refl = (φ ) t) ↑ θ

The monoid of scopes is generated from its singletons. By the time we use a variable, it should be the

only thing in scope. The associated smart constructor computes the thinned representation of variables.

data VaR (k : K) : K where

only : VaR k ([] -, k)
vaR : k← kz → VaR k ⇑ kz

vaR x = only ↑ x

Untyped λ -calculus. We can now give the λ -terms for which all free variables are relevant as follows.

Converting de Bruijn to co-de-Bruijn representation is easy with smart constructors. E.g., compare de

Bruijn terms for the K and S combinators with their co-de-Bruijn form.

data LamR : Onewhere

# : VaR 〈〉 →̇ LamR

app : (LamR ×R LamR) →̇ LamR

λ : ([] -, 〈〉 ⊢ LamR) →̇ LamR

lamR : Lam →̇ (LamR ⇑ )
lamR (# x) = map⇑ # (vaR x)
lamR (f $ s) = map⇑ app (lamR f ,R lamR s)
lamR (λ t) = map⇑ λ ( )R lamR t)

K = λ (λ (# (oe os o′)))
lamR K = λ (oz os ) λ (oz o′ ) # only)) ↑ oz

S = λ (λ (λ (# (oe os o′ o′) $ # (oe os) $ (# (oe os o′) $ # (oe os)))))
lamR S = λ (oz os ) λ (oz os ) λ (oz os )
app (pair (app (pair (# only ↑ oz os o′) (# only ↑ oz o′ os) (czz cs′ c′s)) ↑ oz os o′ os)

(app (pair (# only ↑ oz os o′) (# only ↑ oz o′ os) (czz cs′ c′s)) ↑ oz o′ os os)
(czz cs′ c′s css))))) ↑ oz

Stare bravely! K returns a plainly constant function. Meanwhile, S clearly uses all three inputs: the

function goes left, the argument goes right, and the environment is shared.

8 A Universe of Metasyntaxes-with-Binding

There is nothing specific to the λ -calculus about de Bruijn representation or its co-de-Bruijn counterpart.

We may develop the notions generically for multisorted syntaxes. If the sorts of our syntax are drawn

from set I, then we may characterize terms-with-binding as inhabiting Kinds kz⇒ i, which specify an

extension of the scope with new bindings kz and the sort i for the body of the binder.

record Kind (I : Set) : Setwhere inductive; constructor ⇒
field scope : Bwd (Kind I); sort : I

Kinds offer higher-order abstraction: a bound variable itself has a Kind, being an object sort parametrized

by a scope, where the latter is, as in previous sections, a Bwd list, with K now fixed as Kind I. Object

variables have sorts; meta-variables have Kinds. E.g., in the β -rule, t and s are not object variables like x

(λx. t[x]) s  t[s]

but placeholders, s for some term and t[x] for some term with a parameter which can be and is instantiated,

by x on the left and s on the right. The kind of t is [] -, ([]⇒ 〈〉)⇒ 〈〉.

We may give the syntax of each sort as a function mapping sorts to Descriptions D : I → Desc I.

dataDesc (I : Set) : Set1 where

RecD : Kind I → Desc I; ΣD : (S : Datoid) → (Data S → Desc I) → Desc I

OneD : Desc I; ×D : Desc I → Desc I → Desc I

We may ask for a subterm with a given Kind, so it can bind variables by listing their Kinds left of ⇒.

Descriptions are closed under unit and pairing. We may also ask for terms to be tagged by some sort of

‘constructor’ inhabiting some Datoid, i.e., a set with a decidable equality, given as follows:

dataDecide (X : Set) : Setwhere

yes : X → Decide X

no : (X → Zero) → Decide X

record Datoid : Set1 where

field Data : Set

decide : (x y : Data) → Decide (x == y)

Describing untyped λ -calculus. Define a tag enumeration, then a description.

data LamTag : Setwhere app λ : LamTag

LAMTAG : Datoid

Data LAMTAG = LamTag

decide LAMTAG app app = yes refl

decide LAMTAG app λ = no λ ()
decide LAMTAG λ app = no λ ()
decide LAMTAG λ λ = yes refl

LamD : One → Desc One

LamD 〈〉 = ΣD LAMTAG λ {app → RecD ([]⇒ 〈〉)×D RecD ([]⇒ 〈〉)
; λ → RecD ([] -, ([]⇒ 〈〉)⇒ 〈〉)}

Note that we do not and cannot include a tag or description for the use sites of variables in terms: use of

variables in scope pertains not to the specific syntax, but to the general notion of what it is to be a syntax.

Interpreting Desc as de Bruijn Syntax. Let us give the de Bruijn interpretation of our syntax descrip-

tions. We give meaning to Desc in the traditional manner, interpreting them as strictly positive operators

in some R which gives the semantics to RecD. In recursive positions, the scope grows by the bindings

demanded by the given Kind. At use sites, higher-kinded variables must be instantiated, just like t[x] in

the β -rule example: −→· computes the Description of the spine of actual parameters required.

J | K : ∀ {I} → Desc I → (I → Kind I) → Kind I

J RecD k | R K kz = R (sort k) (kz ++ scope k)
J ΣD S T | R K kz = Σ (Data S) λ s → J T s | R K kz

J OneD | R K kz = One

J S ×D T | R K kz = J S | R K kz× J T | R K kz

−→ : Bwd (Kind I) → Desc I
−→
[] = OneD
−−−→
kz -, k =

−→
kz ×D RecD k

Tying the knot, we find that a term is either a variable instantiated with its spine of actual parameters,

or it is a construct of the syntax for the demanded sort, with subterms in recursive positions.

data Tm (D : I → Desc I) (i : I) kz : Setwhere -- Tm D i : Kind I

#$ : (jz⇒ i)← kz → J
−→
jz | Tm D K kz → Tm D i kz

[ ] : J D i | Tm D K kz → Tm D i kz

Interpreting Desc as co-de-Bruijn Syntax. Now let us interpret Descriptions in co-de-Bruijn style,

enforcing that all variables in scope are relevant, and that binding sites expose vacuity.

J | KR : Desc I → (I → Kind I) → Kind I

J RecD k | R KR = scope k ⊢ R (sort k)
J ΣD S T | R KR = λ kz → Σ (Data S) λ s → J T s | R KR kz

J OneD | R KR = OneR

J S ×D T | R KR = J S | R KR ×R J T | R KR

data TmR (D : I → Desc I) (i : I) : Kind I where

# : (VaR (jz⇒ i)×R J
−→
jz | TmR D KR) →̇ TmR D i

[ ] : J D i | TmR D KR →̇ TmR D i

We can compute co-de-Bruijn terms from de Bruijn terms, generically.

code : Tm D i →̇ (TmR D i ⇑ )
codes : S → J S | Tm D K →̇ (J S | TmR D KR ⇑ )

code ( #$ {jz} x ts) = map⇑ # (vaR x ,R codes
−→
jz ts)

code {D = D} {i = i} [ ts ] = map⇑ [ ] (codes (D i) ts)
codes (RecD k) t = scope k )R code t

codes (ΣD S T) (s , ts) = map⇑ (s , ) (codes (T s) ts)
codes OneD 〈〉 = 〈〉R
codes (S ×D T) (ss , ts) = codes S ss ,R codes T ts

The reverse translation is left as an (easy) exercise in thinning composition for the reader.

9 Hereditary Substitution for Co-de-Bruijn Metasyntax

Let us develop the appropriate notion of substitution for our metasyntax, hereditary in the sense of

Watkins et al. [23]. Substituting a higher-kinded variable requires us further to substitute its parameters.

We shall need a type to represent the fate of each variable in some source scope as we construct a

term in some target scope. I call this type HSub: let us work through it slowly.

record HSub {I} (D : I → Desc I) -- the underlying syntax

(src trg : Bwd (Kind I)) -- source and target scopes

(act : Bwd (Kind I)) -- the active subscope

: Setwhere

constructor ⊑ [ ] :=
field -- to follow

While D, src and trg indicate the task at hand, the extra scope parameter, act, serves a more subtle

purpose: let us see how, presently. The mixfix constructor is intended to suggest that the partition in

the middle splits the source scope into passive and active variables, with different fates, respectively,

thinning into the target scope and actual substitution:

{pass} : Bwd (Kind I)
{passive} : pass⊑ src

{active} : act ⊑ src

passTrg : pass⊑ trg -- passive variables are ‘renamed’

parti : Cover ff passive active -- ff forbids overlap

images : J−→· act | TmR D KR ⇑ trg -- active variables are substituted

It is convenient to store substitution images as a spine, because hereditary substitutions are exactly gen-

erated from spines. Key to the design, however, is to index HSub over the active subscope, as that is

what will conspicuously decrease when a recursive substition is triggered, making termination obvious

— one of my older tricks [18].

Before we see how to perform a substitution, let us think how to weaken one: we certainly push

under binders, jz, extending source and target scopes, crucially preserving the active subscope.

wkHSub : HSub D src trg act → ∀ jz → HSub D (src ++ jz) (trg ++ jz) act

wkHSub (φ ⊑ [ p ] := is) jz = let ! ! p′ = lrCop jz [] in
(φ ++⊑ oi {kz = jz})⊑ [ p ++C p′ ] := thin⇑ (oi++⊑ oe {kz = jz}) is

We extend the partition to make all the bound variables passive and duly grow the thinning on the left.

On the right, co-de-Bruijn representation lets us thin the spine of images at a stroke!

The definition of hereditary substitution is a mutual recursion, terminating because the active scope is

always decreasing: hSub is the main operation on terms; hSubs proceed structurally, following a syntax

description; hered handles the variable case, invokes hSub hereditarily as required.

hSub : HSub D src trg act → TmR D i iz → iz⊑ src → TmR D i ⇑ trg

hSubs : (S : Desc I) → HSub D src trg act →
J S | TmR D KR iz → iz⊑ src → J S | TmR D KR ⇑ trg

hered : (jz⇒ i)← src → HSub D src trg act → J
−→
jz | TmR D KR ⇑ trg → TmR D i ⇑ trg

There is a design choice here: we may either cut the substitution down to fit the support of the term

we are processing, or retain the substitution intact and keep the thinning which embeds the term’s support

in the source scope. The latter makes the termination argument more straightforward, although we are

required to curry a TmR D i ⇑ src as a t : TmR D i iz with a ψ : iz⊑ src. Our first move is to refine the

substitution’s partition by ψ to check whether any of the variables in the term’s support is actively being

substituted. If not, we may simply thin t, with no further traversal.

hSub h@ (φ ⊑ [ p′ ] := is) t ψ with subCop ψ p′

hSub h@ (φ ⊑ [ p′ ] := is) t ψ | , [] , , , ψ0 , ψ1 , p with allLeft p

hSub h@ (φ ⊑ [ p′ ] := is) t ψ | , [] , , , ψ0 , ψ1 , p | refl = t ↑ (ψ0 # φ)

The [] pattern matches the active part of the support, with allLeft the lemma that the passive part must be

the whole support if the active part is empty. If, on the other hand, there are still active variables to find,

we must keep hunting, in the knowledge that we have real work to do.

hSub {D = D} {i = i} h@ [ ts ] ψ | = map⇑ [ ] (hSubs (D i) h ts ψ)

If we find a node from our syntax, we proceed structurally:

hSubs (RecD (jz⇒ i)) h (θ ) t) ψ = jz )R hSub (wkHSub h jz) t (ψ ++⊑ θ)
hSubs (ΣD S T) h (s , ts) ψ = map⇑ (s , ) (hSubs (T s) h ts ψ)
hSubs OneD h = 〈〉R
hSubs (S×D T) h (pair (s ↑ θ ) (t ↑ φ) ) ψ = hSubs S h s (θ # ψ) ,R hSubs T h t (φ # ψ)

Meanwhile, for a variable with spine attached, we substitute the spine then proceed hereditarily.

hSub h@ (# {jz} (pair (only ↑ x) (ss ↑ θ ) c)) ψ | = hered (x # ψ) h (hSubs
−→
jz h ss (θ # ψ))

If the variable we seek is not the top one in the source context, we throw the top variable, passive or

active, out of the substitution and keep looking.

hered (x o′) (φ ⊑ [ p cs′ ] := is) ss = hered x (oi o′ # φ ⊑ [ p ] := is) ss

hered (x o′) (φ ⊑ [ p c′s ] := is) ss = hered x (φ ⊑ [ p ] := outlR is) ss

We must rule out the possibility that any variable is both active and passive.

hered ( ⊑ [ css {both = ()} ] := )

Now we have found our variable, and it is either passive (in which case we attach the spine). . .

hered (x os) (φ ⊑ [ p cs′ ] := ) ss = map⇑ # (vaR (oe os # φ) ,R ss)

. . . or active, in which case we substitute hereditarily.

hered {trg = trg} {act = ( -, (jz⇒ i))} (x os) ( ⊑ [ p c′s ] := is) ss

with outrR is | lrCop trg jz

... | (ψ ) t) ↑ θ | ! ! p′ = hSub {act = jz} (oi ⊑ [ p′ ] := ss) t (θ ++⊑ ψ)

As you can see, the target scope becomes passive, the bound variables of the substitution image become

active, and the spine becomes the substitution for the active variables. The new active scope is visibly a

substructure of the old active scope, so hereditary substitution is structurally recursive!

10 Discussion

We have a universe of syntaxes with metavariables and binding, where the Description of a syntax is

interpreted as the co-de-Bruijn terms, ensuring intrinsically that unused variables are discarded not at the

latest opportunity (as in de Bruijn terms), nor at an arbitrary opportunity (as in one of Bird and Paterson’s

variants [8], or with Hendriks and van Oostrom’s ‘adbmal’ operator [14], both of which reduce the labour

of shifting at the cost of nontrivial α-equivalence), but at the earliest opportunity. Hereditary substitution

exploits usage information to stop when there is nothing to substitute, shifts without traversal, and is,

moreover, structurally recursive on the active scope.

Co-de-Bruijn representation is even less suited to human comprehension than de Bruijn syntax, but

its informative precision makes it all the more useful for machines. Dependency checking is direct, so

syntactic forms like vacuous functions or η-redexes are easy to spot.

It remains to be seen whether co-de-Bruijn representation will lead to more efficient implementations

of normalization and of metavariable instantiation. The technique may be readily combined with repre-

senting terms as trees whose top-level leaves are variable uses and top-level nodes are just those (now

easily detected) where paths to variables split: edges in the tree are closed one-hole contexts, jumped

over in constant time [10].

I see two high-level directions emerging from this work. Firstly, the generic treatment of syntax

with metavariables opens the way to the generic treatment of metatheory. Even without moving from

scope-safe to type-safe term representations, we can generate the inductive relations we use to define

notions such as reduction and type synthesis in a universe, then seek to capture good properties (e.g.,

stability under substitution, leading to type soundness) by construction. Co-de-Bruijn representations

make it easy to capture properties such as variable non-occurrence in the syntax of formulæ, and might

also serve as the target term representation for algorithms extracted generically from the rules.

Secondly, more broadly, this work gives further evidence for a way of programming with strong

invariants and redundant but convenient information caches without fear of bugs arising from inconsis-

tency. We should put the programmer in charge! Dependent types should let us take control of data

representations and optimise them to support key operations, but with the invariants clearly expressed in

code and actively supporting program synthesis.

Only a fool would attempt to enforce the co-de-Bruijn invariants without support from a typechecker,

so naturally I have done so: using Haskell’s Integer for bit vectors (making -1 the identity of the

unscoped thinning monoid), I implemented a dependent type system, just for fun. It was Hell’s delight,

even with the Agda version to follow. I was sixteen again.

